Appendix H

Sample DOE 2.1D Input and Output for Analysis



BUILDING ENERGY ANALYSIS PROGRAM

DEVELOPED BY

LAWRENCE BERKELEY LABORATORY/UNIVERSITY OF CALIFORNIA

WITH MAJOR SUPPORT FROM

UNITED STATES DEPARTMENT OF ENERGY
ASSISTANT SECRETARY FOR CONSERVATION AND RENEWABLE ENERGY
OFFICE OF BUILDINGS AND COMMUNITY SYSTEMS
BUILDING SYSTEMS DIVISION

LBL RELEASE FEB 1989 version : 2.1D-014

```
INPUT LOADS ..
TITLE    LINE-1 *Single Family Residence Base Case *
         LINE-2 *---New 1500 Square Ft Prototype---*
         LINE-3 * Frame Construction for Miami *
         LINE-4 * REFERENCE CASE *
         LINE-5 * D. Parker/ Nov. 18, 1992 * ..

         DIAGNOSTIC          ERRORS, NO-ECHO ..
         RUN-PERIOD          JAN 1 1989 THRU DEC 31 1989 ..
         BUILDING-LOCATION   LAT=25.5  LON=80.2  T-Z=5
                             AZIMUTH=0  DAYLIGHT-SAVINGS= YES ..
         LOADS-REPORT        SUMMARY=(LS-B) ..

                $---SCHEDULE----$

SCH-1    =DAY-SCHEDULE    (1) (.32) (2) (.30) (3,4) (.29) (5) (.31)
                          (6) (.46) (7) (.39) (8) (.44) (9) (.46) (10) (.56)
                          (11) (.63) (12) (.59) (13) (.55) (14) (.52)
                          (15) (.57) (16) (.48) (17) (.56) (18) (.77)
                          (19) (.82) (20) (.84) (21) (.99) (22) (1.00)
                          (23) (.69) (24) (.47) ..
INT-LDS-1 =SCHEDULE       THRU DEC 31 (ALL) SCH-1 ..

                $---MATERIALS---$

STUD-1     =MAT      TH=.2917 COND=.0667 DENS=32  S-H=.33 $2 X 4 STUD$ ..
DRYWALL-1  =MAT      TH=.0417 COND=.0925 DENS=50  S-H=.26 $1/2 IN DRYWALL$ ..
WALL-INS-1=MAT       TH=.2618 COND=.0238 DENS=6   S-H=.2  $R-11 INSULATION$ ..
WALL-INS-2=MAT       TH=.0833 COND=.0238 DENS=6   S-H=.2  $R-3 INSULATION$ ..
SHEATH-1   =MAT      TH=.0417 COND=.0342 DENS=22  S-H=.31 $1/2 IN SHEATHING$ ..
AL-SIDE-1  =MAT      TH=.0104 COND=.0171 DENS=170 S-H=.29 $ALUM SIDING$ ..
BLOCK-1    =MAT      TH=.6667 COND=.6060 DENS=69  S-H=.2  $8" CON. BLOCK$ ..
STUCCO-1   =MAT      TH=.0833 COND=.4167 DENS=166 S-H= .2 $STUCCO SIDING$ ..
AS-SHG-1   =MAT      TH=.0208 COND=.0473 DENS=70  S-H=.30 $ASPHALT SHINGLE$ ..
PLYW-1     =MAT      TH=.0417 COND=.0667 DENS=34  S-H=.29 $1/2 IN PLYWOOD$ ..
CEIL-INS-1=MAT       TH=.4522 COND=.0238 DENS=6   S-H=.2  $R-19 INSULATION$ ..
AT-AIR-1   =MAT      RES=3.1                              $ATTIC AIR SPACE$ ..
EXP-POLY-1=MAT       TH=.0833 COND=.0167 DENS=2.2 S-H=.29 $1 IN POLYSTYRENE$ ..
CONCRETE-1=MAT       TH=.3333 COND=.7576 DENS=140 S-H=.2  $4 IN CONCRETE$ ..
CARP/PAD-1=MAT       RES=2.08                             $CARPET AND PAD$ ..

                $---GLAZINGS---$

GT-WIN-1   =GLASS-TYPE     PANES=1  S-C=.6 GLASS-CONDUCTANCE = 1.1 ..
GT-WIN-2   = GLASS-TYPE    PANES=1  S-C=.6  GLASS-CONDUCTANCE = 1.1 ..

                $---CONSTRUCTIONS---$

LAY-1        =LAYERS       MAT=(AL-SIDE-1, SHEATH-1, WALL-INS-1, DRYWALL-1) ..
INS-WL-1     =CONS         LAYERS=LAY-1  ROUGHNESS=6  ABS = 0.6 ..

LAY-2        =LAYERS       MAT=(AL-SIDE-1, SHEATH-1, STUD-1, DRYWALL-1) ..
STUD-WL-1    =CONS         LAYERS=LAY-2  ROUGHNESS=6  ABS = 0.6 ..

LAY-3        =LAYERS       MAT=(AS-SHG-1, PLYW-1, AT-AIR-1, CEIL-INS-1,
                                DRYWALL-1)  I-F-R=.61 ..
INS-RF-1     =CONS         LAYERS=LAY-3  ABS=.8 ..

LAY-4        =LAYERS       MAT=(AS-SHG-1, PLYW-1, STUD-1, AT-AIR-1, STUD-1,
                                DRYWALL-1)  I-F-R=.92 ..
STUD-RF-1    =CONS         LAYERS=LAY-4   ABS=.8 ..

LAY-5        =LAYERS       MAT=(EXP-POLY-1, CONCRETE-1, CARP/PAD-1)
                                I-F-R=.92 ..
SLAB-1       =CONS         LAYERS=LAY-5 ..
DR-1         =CONS         U=.6  ABS=.6  ROUGHNESS=4 ..

               $---BUILDING SHADE---$

       BUILDING-SHADE      X=0  Y=0  Z=8  H=2  W=50  $FRONT OVERHANGS$
                           AZ=180  TILT=180  ..
       BUILDING-SHADE      X=0  Y=30  Z=8  H=2  W=50  $REAR OVERHANGS$
                           AZ=180  TILT=0 ..

               $----SPACE DESCRIPTIONS----$

COND-1 =SPACE-CONDITIONS   SOURCE-SCHEDULE=INT-LDS-1
                           SOURCE-TYPE=PROCESS  SOURCE-BTU/HR=4304
                           INF-METHOD=AIR-CHANGE  FLOOR-WEIGHT = 0
                           AIR-CHANGES/HR = 0.40 ..
```

```
  HOUSE-1     =SPACE          A=1500 V=12000  S-C=COND-1 ..

  FRONT-WL-1=E-W              H=8  W=43.75  AZ=180  CONS=INS-WL-1 ..
  WIN-1       =WI             H=4.5  W=15.56  G-T=GT-WIN-1  X=18  Y=0 ..
  DOOR-1      =DOOR           H=6.67  W=3  CONS=DR-1  X=24.5 ..

  FRONT-WL-2=E-W              H=8  W=6.25 AZ=180  CONS=STUD-RF-1 ..

  RIGHT-WL-1=E-W              H=8  W=26.25  AZ=90  X=50  Y=0  CONS=INS-WL-1 ..
  WIN-2       =WI             H=4.5  W= 9.33 G-T= GT-WIN-2  X= 50 Y= 14 ..

  RIGHT-WL-2=E-W              H=8  W=3.75 AZ=90  X=50  Y=0  CONS=STUD-WL-1 ..

  REAR-WL-1 =E-W              LIKE FRONT-WL-1  W=43.75  X=50  Y=30  AZ=0 ..
  WIN-3       =WI             H=4.5  W=15.56  X=21  Y=0  G-T=GT-WIN-1
                              RIGHT-FIN-A = 10 RIGHT-FIN-H= 8 RIGHT-FIN-D = 24 ..

  REAR-WL-2 =E-W              LIKE FRONT-WL-2  W=6.25  X=50  Y=30  AZ=0 ..

  LEFT-WL-1 =E-W              LIKE RIGHT-WL-1  X=0  Y=30  AZ=270 ..
  WIN-4       =WI             H=4.5  W=9.33  G-T=GT-WIN-2  X=0 Y= 14  ..

  LEFT-WL-2 =E-W              LIKE RIGHT-WL-2  X=0  Y=30  AZ=270 ..

  FRONT-RF-1=ROOF             H=15  W=45  Z=8  AZ=180  TILT=22.62
                              CONS=INS-RF-1 ..

  FRONT-RF-2=ROOF             H=15  W=5  Z=8  AZ=180  TILT=22.62
                              CONS=STUD-RF-1 ..
  REAR-RF-1=ROOF              LIKE FRONT-RF-1  X=50  Y=30  AZ=0 ..

  REAR-RF-2=ROOF             LIKE FRONT-RF-2  X=50  Y=30  AZ=0 ..

            U-F               CONS=SLAB-1  A=160  $PERIMETER AREA$ ..

  END ..
  COMPUTE LOADS ..
  INPUT SYSTEMS ..
              SYSTEMS-REPORT      SUMMARY=(SS-J, SS-H) ..


                    $---SCHEDULES---$

  HEAT-1      =SCHEDULE           THRU MAR 31 (WD) (1,7) (68) (8,17) (69)
                                  (18,23) (70) (24) (68)
                                  (WEH) (1,8) (68) (9,24) (70)
                                  THRU NOV 30 (ALL) (1,24) (55)
                                  THRU DEC 31 (WD) (1,7) (68) (8,17) (69)
                                   (18,23) (70) (24) (68)
                                  (WEH) (1,8) (68) (9,24) (70) ..
  COOL-1      =SCHEDULE           THRU JAN 31 (ALL) (1,24) (150)
                                  THRU NOV 30 (WD) (1,7) (78) (8,17) (80)
                                  (18,24) (78)
                                  (WEH) (1,24) (78)
                                  THRU DEC 31 (ALL) (1,24) (150) ..
  VENT-1      =SCHEDULE           THRU DEC 31 (ALL) (1,24) (1) ..
  VENTING     =SCHEDULE           THRU DEC 31 (ALL) (1,24) (68) ..

  HOUSE-1     =ZONE               DESIGN-HEAT-T=70  DESIGN-COOL-T=78
                                  ZONE-TYPE=CONDITIONED
                                  ASSIGNED-CFM=1200
                                  THERMOSTAT-TYPE=TWO-POSITION
                                  HEAT-TEMP-SCH=HEAT-1
                                  COOL-TEMP-SCH=COOL-1 ..

                  $--AIR CONDITIONER AND HEATING SYSTEM PARAMETERS--$

  SYS-1       =SYSTEM             SYSTEM-TYPE=RESYS  ZONE-NAMES=(HOUSE-1)
                                  MAX-SUPPLY-T=140  MIN-SUPPLY-T=50
                                  COOLING-CAPACITY=36000
                                  COOL-SH-CAP=22500  COOLING-EIR= .3333
                                  HEATING-CAPACITY=-40000
                                  HEAT-SOURCE= ELECTRIC  FURNACE-AUX=100
                                  NATURAL-VENT-AC=10  NATURAL-VENT-SCH=VENT-1
                                  VENT-TEMP-SCH= VENTING ..


  END ..
  COMPUTE SYSTEMS ..
```

L D L P R O C E S S O R I N P U T D A T A

12/11/1992 18:03:09 LDL RUN 1

```
*   2 *  TITLE    LINE-1 *Single Family Residence Base Case *
*   3 *           LINE-2 *---New 1500 Square Ft Prototype---*
*   4 *           LINE-3 * Frame Construction for Miami *
*   5 *           LINE-4 * REFERENCE CASE *
*   6 *           LINE-5 * D. Parker/ Nov. 18, 1992 * ..
*   7 *
*   8 *           DIAGNOSTIC          ERRORS, NO-ECHO ..
* 118 * COMPUTE LOADS ..
* 119 * INPUT SYSTEMS ..
```

S D L P R O C E S S O R I N P U T D A T A

12/11/1992 18:03:09 SDL RUN 1

```
* 160 * COMPUTE SYSTEMS ..
* 161 * STOP ..
```

SPACE HOUSE-1

MULTIPLIER 1.0 FLOOR MULTIPLIER 1.0

FLOOR AREA 1500 SQFT 139 M2
VOLUME 12000 CUFT 340 M3

	COOLING LOAD		HEATING LOAD	
TIME	APR 2 3PM		DEC 16 7AM	
DRY-BULB TEMP	89F	32C	35F	2C
WET-BULB TEMP	73F	23C	32F	0C

	SENSIBLE (KBTU/H)	SENSIBLE (KW)	LATENT (KBTU/H)	LATENT (KW)	SENSIBLE (KBTU/H)	SENSIBLE (KW)
WALLS	1.206	0.353	0.000	0.000	-2.340	-0.685
ROOFS	1.432	0.419	0.000	0.000	-1.909	-0.559
GLASS CONDUCTION	3.169	0.928	0.000	0.000	-6.612	-1.936
GLASS SOLAR	4.614	1.351	0.000	0.000	0.365	0.107
DOOR	0.163	0.048	0.000	0.000	-0.358	-0.105
INTERNAL SURFACES	0.000	0.000	0.000	0.000	0.000	0.000
UNDERGROUND SURFACES	0.094	0.028	0.000	0.000	0.104	0.030
OCCUPANTS TO SPACE	0.000	0.000	0.000	0.000	0.000	0.000
LIGHT TO SPACE	0.000	0.000	0.000	0.000	0.000	0.000
EQUIPMENT TO SPACE	0.000	0.000	0.000	0.000	0.000	0.000
PROCESS TO SPACE	2.089	0.612	0.000	0.000	1.484	0.435
INFILTRATION	4.706	1.378	5.236	1.533	-3.514	-1.029
TOTAL	17.473	5.117	5.236	1.533	-12.779	-3.743
TOTAL LOAD	22.709 KBTU/H		6.651 KW		-12.779 KBTU/H	-3.743 KW
TOTAL LOAD / AREA	15.14BTU/H.SQFT		47.726 W / M2		8.519BTU/H.SQFT	26.857 W / M2

```
**********************************************************************
*                                                                    *
*  NOTE  1)THE ABOVE LOADS EXCLUDE OUTSIDE VENTILATION AIR           *
*  ----    LOADS                                                     *
*        2)TIMES GIVEN IN STANDARD TIME FOR THE LOCATION             *
*          IN CONSIDERATION                                          *
*                                                                    *
**********************************************************************
```

SYSTEM NAME	ALTITUDE MULTIPLIER
SYS-1	1.000

SUPPLY FAN (CFM)	ELEC (KW)	DELTA-T (F)	RETURN FAN (CFM)	ELEC (KW)	DELTA-T (F)	OUTSIDE AIR RATIO	COOLING CAPACITY (KBTU/HR)	SENSIBLE (SHR)	HEATING CAPACITY (KBTU/HR)	COOLING EIR (BTU/BTU)	HEATING EIR (BTU/BTU)
1200.	0.154	0.4	0.	0.000	0.0	0.000	36.000	0.625	-40.000	0.33	0.37

ZONE NAME	SUPPLY FLOW	EXHAUST FLOW	FAN (KW)	MINIMUM FLOW RATIO	OUTSIDE AIR FLOW	COOLING CAPACITY (KBTU/HR)	SENSIBLE (SHR)	EXTRACTION RATE (KBTU/HR)	HEATING CAPACITY (KBTU/HR)	ADDITION RATE (KBTU/HR)	MULTIPLIER
HOUSE-1	1200.	0.	0.000	1.000	0.	0.00	0.00	30.30	0.00	-40.47	1.0

	-----COOLING-----				---HEATING---			DAY COOLING PEAK			
	MAY 3				DEC 16			AUG 4			
HOUR	HOURLY COOLING LOAD (KBTU)	SENSIBLE HEAT RATIO	DRY-BULB TEMP	WET-BULB TEMP	HOURLY HEATING LOAD (KBTU)	DRY-BULB TEMP	WET-BULB TEMP	HOURLY COOLING LOAD (KBTU)	SENSIBLE HEAT RATIO	DRY-BULB TEMP	WET-BULB TEMP
1	4.157	0.596	80.F	77.F	-8.016	41.F	37.F	5.155	0.632	82.F	75.F
2	4.096	0.560	80.F	77.F	-8.759	40.F	37.F	4.576	0.643	82.F	75.F
3	3.171	0.541	78.F	77.F	-9.082	38.F	36.F	4.265	0.645	82.F	75.F
4	0.000	0.000	77.F	76.F	-9.353	37.F	35.F	4.324	0.619	82.F	76.F
5	0.000	0.000	77.F	76.F	-10.201	36.F	34.F	4.818	0.617	82.F	76.F
6	0.000	0.000	76.F	75.F	-10.837	36.F	33.F	4.280	0.630	82.F	77.F
7	0.000	0.000	77.F	76.F	-11.768	35.F	32.F	1.805	0.634	82.F	76.F
8	3.625	0.621	80.F	77.F	-9.163	39.F	34.F	6.380	0.652	85.F	76.F
9	8.557	0.639	83.F	78.F	-8.210	42.F	36.F	8.102	0.651	83.F	76.F
10	10.117	0.725	84.F	77.F	-4.259	46.F	38.F	10.537	0.722	87.F	77.F
11	12.312	0.708	86.F	77.F	-1.520	50.F	40.F	12.816	0.723	88.F	77.F
12	12.690	0.753	89.F	77.F	0.000	53.F	41.F	14.021	0.695	88.F	78.F
13	13.944	0.739	90.F	77.F	0.000	57.F	42.F	13.977	0.721	88.F	76.F
14	14.340	0.744	88.F	76.F	0.000	58.F	43.F	14.685	0.733	89.F	77.F
15	14.755	0.759	88.F	76.F	0.000	59.F	44.F	15.609	0.733	89.F	77.F
16	16.172	0.753	88.F	76.F	0.000	60.F	45.F	13.726	0.758	85.F	76.F
17	20.184	0.748	86.F	77.F	0.000	57.F	43.F	19.773	0.754	87.F	77.F
18	16.368	0.747	85.F	76.F	-0.741	53.F	42.F	16.421	0.709	85.F	78.F
19	11.923	0.748	82.F	76.F	-2.054	50.F	40.F	14.502	0.665	84.F	77.F
20	10.448	0.713	81.F	76.F	-3.237	49.F	41.F	12.689	0.656	83.F	77.F
21	8.817	0.691	79.F	75.F	-3.920	48.F	42.F	11.163	0.647	82.F	77.F
22	6.924	0.665	79.F	75.F	-4.690	47.F	43.F	9.097	0.637	82.F	77.F
23	0.000	0.000	76.F	72.F	-5.452	47.F	42.F	6.904	0.564	79.F	78.F
24	0.000	0.000	73.F	70.F	-5.972	47.F	42.F	4.000	0.700	79.F	74.F
SUM								233.624			
MAX	20.184				-11.768						

	--FAN ELEC---		--FUEL HEAT--		--FUEL COOL--		-ELEC HEAT--		-ELEC COOL--	
ONTH	FAN ENERGY (KWH)	MAXIMUM FAN LOAD (KW)	GAS OIL ENERGY (MBTU)	MAXIMUM GAS OIL LOAD (KBTU/HR)	GAS OIL ENERGY (MBTU)	MAXIMUM GAS OIL LOAD (KBTU/HR)	ELECTRIC ENERGY (KWH)	MAXIMUM ELECTRIC LOAD (KW)	ELECTRIC ENERGY (KWH)	MAXIMUM ELECTRIC LOAD (KW)
AN	2.	0.042	0.000	0.000	0.000	0.000	149.	3.208	11.	0.050
EB	7.	0.072	0.000	0.000	0.000	0.000	116.	2.406	124.	1.463
AR	7.	0.077	0.000	0.000	0.000	0.000	8.	0.947	166.	1.502
PR	12.	0.087	0.000	0.000	0.000	0.000	0.	0.000	268.	1.747
AY	17.	0.102	0.000	0.000	0.000	0.000	0.	0.000	406.	1.973
JUN	23.	0.099	0.000	0.000	0.000	0.000	0.	0.000	547.	1.912
JUL	25.	0.101	0.000	0.000	0.000	0.000	0.	0.000	573.	1.908
AUG	28.	0.105	0.000	0.000	0.000	0.000	0.	0.000	669.	2.014
SEP	25.	0.095	0.000	0.000	0.000	0.000	0.	0.000	599.	1.830
OCT	16.	0.099	0.000	0.000	0.000	0.000	0.	0.000	380.	1.834
NOV	9.	0.075	0.000	0.000	0.000	0.000	0.	0.000	203.	1.470
DEC	3.	0.045	0.000	0.000	0.000	0.000	203.	3.447	13.	0.050
TOTAL	175.		0.000		0.000		476.		3960.	
MAX		0.105		0.000		0.000		3.447		2.014